Παράδειγμα #7
ΑΡΙΘΜΗΤΙΚΗ ΟΛΟΚΛΗΡΩΣΗ
ΕΠΙΜΕΛΕΙΑ: Ν. Βασιλειάδης

<u>Άσκηση 1</u>

Έστω ότι σε ένα δοκάρι μήκους L=40 m ασκείται μεταβαλλόμενο φορτίο:

$$f(z) = 100\left(\frac{z}{3+z}\right)e^{\frac{-2z}{15}} \text{ (N/m)}$$

Να υπολογισθεί η συνολική δύναμη και το σημείο που ασκείται, εφαρμόζοντας τις μεθόδους αριθμητικής ολοκλήρωσης τραπεζίου και του 1^{ου} και 2^{ου} κανόνα Simpson.

Απάντηση:

Από την μηχανική είναι γνωστό ότι συνολική δύναμη υπολογίζεται ως $F = \int_0^L f(z)dz$

ενώ η θέση στην οποία ασκείται υπολογίζεται ως: $z_F = \dfrac{1}{F}\int_0^L zf(z)\,dz$

Η συνολική δύναμη και το σημείου που ασκείται μέσω Mathematica υπολογίζονται ως:

```
L=40;
f=100*(z/(3+z))*Exp[-2*z/15];
F=N[Integrate[f,{z,0,L}],5]
zF=N[Integrate[z*f,{z,0,L}]/F,5]
```

Από τον παραπάνω κώδικα προκύπτει ότι η συνολική δύναμη είναι F=432.25 N και ασκείται στο σημείο z_F=9.6154.

Ο κώδικας ο οποίος υπολογίζει οποιοδήποτε μονό ολοκλήρωμα με τους κανόνες τραπεζίου, 1^o και 2^o κανόνα Simpson δίνεται παρακάτω:

```
program NewtonCotes1D
implicit none
integer::i,n,m,IntRule
real*8::a,b,h,x,w,NumInt

!Definition of numerical integration
a=0.        !Lower integral bound
b=40.       !Upper integral bound
m=1         !Number of integration intervals
IntRule=1   !1:Trapezoidal Rule
            !2:1st Simpson rule
            !3:2nd Simpson rule
If (IntRule==1) n=m+1
If (IntRule==2) n=3+(m-1)*2
If (IntRule==3) n=4+(m-1)*3
```

```
h=(b-a)/Dble(n-1.)

NumInt=0.
do i=1,n
    x=a+(i-1.)*h
    w=weight(i,n)
    NumInt=NumInt+f(x)*w
enddo
If (IntRule==1) NumInt=NumInt*(h/2.)
If (IntRule==2) NumInt=NumInt*(h/3.)
If (IntRule==3) NumInt=NumInt*(3*h/8.)

write(*,"(A,F15.6)") "Numerical integration result:",NumInt

contains

!Function to be integrated
real*8 function f(x)
real*8,intent(in)::x

f=100.*(x/(3.+x))*Exp(-2.*x/15.)

end function f

real*8 function weight(indx,n)
integer,intent(in)::indx,n
if (IntRule==1) then
    if (indx==1 .or. indx==n) then
        weight=1.
    else
        weight=2.
    endif
elseif (IntRule==2) then
    if (indx==1 .or. indx==n) then
        weight=1.
    elseif (mod(indx,2)==0) then
        weight=4.
    else
        weight=2.
    endif
elseif (IntRule==3) then
    if (indx==1 .or. indx==n) then
        weight=1.
    elseif (mod(indx-1,3)==0) then
        weight=2.
    else
        weight=3.
    endif
endif
end function weight

end
```

Εκτελώντας τον παραπάνω κώδικα για τους τρεις κανόνες ολοκλήρωσης Newton-Cotes και αυξάνοντας διαδοχικά τα υποδιαστήματα προκύπτει για την δύναμη:

Υποδιαστήματα	Συνολική δύναμη $F\,[N]$		
	Τραπέζιο	1ος Simpson	2ος Simpson
1	8.9822	164.12	247.72
2	125.33	334.33	374.79
5	321.07	419.75	425.88
10	395.08	430.62	431.48
100	431.81	432.25	432.25

Για τον υπολογισμό του $\int_0^L zf(z)\,dz$ η μόνη αλλαγή που απαιτείται στον κώδικα είναι:

```
f=x*100.*(x/(3.+x))*Exp(-2.*x/15.)
```

Έχοντας την τιμή του ολοκληρώματος $\int_0^L zf(z)\,dz$ και διαιρώντας με τα αποτελέσματα για την συνολική δύναμη από τον παραπάνω πίνακα προκύπτει το σημείο στο οποίο ασκείται η δύναμη:

Υποδιαστήματα	Σημείο άσκησης δύναμης $z_F\,[m]$		
	Τραπέζιο	1ος Simpson	2ος Simpson
1	40.000	20.365	15.648
2	20.717	12.667	11.284
5	12.379	9.9740	9.7964
10	10.463	9.6620	9.6374
100	9.6252	9.6154	9.6154

Άσκηση 2

Να εφαρμοστούν οι μέθοδοι αριθμητικής ολοκλήρωσης τραπεζίου και του 1ου και 2ου κανόνα Simpson για να υπολογισθούν τα ολοκληρώματα:

$$\text{α)}\quad I_1 = \int_0^1\int_0^x \sqrt{x+y}\,dydx \qquad \text{β)}\quad I_2 = \int_0^{2\pi}\int_0^{\pi}\int_0^{\pi/2} (x+y+z)\sin(x+y+z)\,dydxdz$$

Απάντηση:

2α. Διπλό ολοκλήρωμα

Για τον αριθμητικό υπολογισμό του διπλού ολοκληρώματος είναι αναγκαίος ο μετασχηματισμός του μέσω της νέας μεταβλητής:

$$s=\frac{y-(b+a)}{(b-a)}=\frac{y-x}{x}\Rightarrow y=s(x+1),\quad dy=xds$$

Το ισοδύναμο διπλό ολοκλήρωμα είναι:

$$I_1=\int_0^1\int_0^x\sqrt{x+y}\,dydx=\int_0^1\int_{-1}^0 x\sqrt{x(s+2)}\,dsdx$$

Ο υπολογισμός του παραπάνω ολοκληρώματος μέσω Mathematica γίνεται ως:

```
NumberForm[NIntegrate[x*Sqrt[x*(s+2)],{s,-1,0},{x,0,1}],5]
```

Από τον οποίο προκύπτει η τιμή του ολοκληρώματος $I_1=0.48758$.

Ο κώδικας σε Fortran που υπολογίζει οποιοδήποτε διπλό ολοκλήρωμα με τους κανόνες τραπεζίου, 1^o και 2^o κανόνα Simpson δίνεται παρακάτω:

```
program NewtonCotes2D
implicit none
integer::i,j,n1,n2,m1,m2,IntRule
real*8::a1,b1,a2,b2,h1,h2,x1,x2,w1,w2,NumInt

!Definition of numerical integration
a1=0.; b1=1.      !Integral bounds in 1st direction
a2=-1.; b2=0.     !Integral bounds in 2nd direction
m1=1              !No. of intervals in 1st direction
m2=1              !No. of intervals in 2nd direction
IntRule=1         !1:Trapezoidal Rule
                  !2:1st Simpson rule
                  !3:2nd Simpson rule

if (IntRule==1) then
  n1=m1+1
  n2=m2+1
endif
if (IntRule==2) then
  n1=3+(m1-1)*2
  n2=3+(m2-1)*2
endif
if (IntRule==3) then
  n1=4+(m1-1)*3
  n2=4+(m2-1)*3
endif
h1=(b1-a1)/Dble(n1-1.)
h2=(b2-a2)/Dble(n2-1.)

NumInt=0.
do i=1,n1
    do j=1,n2
        x1=a1+(i-1.)*h1
        x2=a2+(j-1.)*h2
        !write(*,*) x1,x2
        w1=weight(i,n1)
```

```
        w2=weight(j,n2)
        NumInt=NumInt+f(x1,x2)*w1*w2
    enddo
enddo
If (IntRule==1) NumInt=NumInt*(h1/2.)*(h2/2.)
If (IntRule==2) NumInt=NumInt*(h1/3.)*(h2/3.)
If (IntRule==3) NumInt=NumInt*(3*h1/8.)*(3*h2/8.)

write(*,"(A,F15.6)") "Numerical integration result:",NumInt

contains

!Function to be integrated
real*8 function f(x1,x2)
real*8,intent(in)::x1,x2

f=x1*Sqrt(x1*(x2+2.))

end function f

real*8 function weight(indx,n)
integer,intent(in)::indx,n
if (IntRule==1) then
    if (indx==1 .or. indx==n) then
        weight=1.
    else
        weight=2.
    endif
elseif (IntRule==2) then
    if (indx==1 .or. indx==n) then
        weight=1.
    elseif (mod(indx,2)==0) then
        weight=4.
    else
        weight=2.
    endif
elseif (IntRule==3) then
    if (indx==1 .or. indx==n) then
        weight=1.
    elseif (mod(indx-1,3)==0) then
        weight=2.
    else
        weight=3.
    endif
endif
end function weight

end
```

Εκτελώντας τον παραπάνω κώδικα για του τρεις κανόνες ολοκλήρωσης Newton-Cotes και αυξάνοντας διαδοχικά τα υποδιαστήματα προκύπτουν τα παρακάτω:

Υποδιαστήματα	Τραπέζιο	1ος Simpson	2ος Simpson
1×1	0.60355	0.49043	0.48914
2×2	0.51893	0.48811	0.48787
5×5	0.49292	0.48764	0.48761
10×10	0.48896	0.48759	0.48759
100×100	0.48760	0.48758	0.48758

2β. Τριπλό ολοκλήρωμα

Ο υπολογισμός του ζητούμενου τριπλού ολοκληρώματος μέσω Mathematica γίνεται ως:

```
NumberForm[NIntegrate[(x+y+z)*Sin[x+y+z], {y,0,Pi/2.},{x,0,Pi},{z,0,
2*Pi}],5]
```

Από τον παραπάνω κώδικα προκύπτει η τιμή του ολοκληρώματος $I_2 = 12.566$.

Ο κώδικας σε Fortran που υπολογίζει οποιοδήποτε διπλό ολοκλήρωμα με τους κανόνες τραπεζίου, 1ο και 2ο κανόνα Simpson δίνεται παρακάτω:

```
program NewtonCotes3D
implicit none
integer::i,j,k,n1,n2,n3,m1,m2,m3,IntRule
real*8::Pi,a1,b1,a2,b2,a3,b3,h1,h2,h3,x1,x2,x3,w1,w2,w3,NumInt

!Definition of numerical integration
Pi=Acos(-1.)
a1=0.; b1=Pi       !Integral bounds in 1st direction
a2=0.; b2=Pi/2.    !Integral bounds in 2nd direction
a3=0.; b3=2*Pi     !Integral bounds in 3rd direction
m1=1               !No. of intervals in 1st direction
m2=1               !No. of intervals in 2nd direction
m3=1               !No. of intervals in 3rd direction
IntRule=1          !1:Trapezoidal Rule
                   !2:1st Simpson rule
                   !3:2nd Simpson rule

if (IntRule==1) then
  n1=m1+1
  n2=m2+1
  n3=m3+1
endif
if (IntRule==2) then
  n1=3+(m1-1)*2
  n2=3+(m2-1)*2
  n3=3+(m3-1)*2
endif
if (IntRule==3) then
  n1=4+(m1-1)*3
  n2=4+(m2-1)*3
  n3=4+(m3-1)*3
endif
```

```
h1=(b1-a1)/Dble(n1-1.)
h2=(b2-a2)/Dble(n2-1.)
h3=(b3-a3)/Dble(n3-1.)
NumInt=0.
do i=1,n1
    do j=1,n2
        do k=1,n3
            x1=a1+(i-1.)*h1
            x2=a2+(j-1.)*h2
            x3=a3+(k-1.)*h3
            w1=weight(i,n1)
            w2=weight(j,n2)
            w3=weight(k,n3)
            NumInt=NumInt+f(x1,x2,x3)*w1*w2*w3
        enddo
    enddo
enddo
If (IntRule==1) NumInt=NumInt*(h1/2.)*(h2/2.)*(h3/2.)
If (IntRule==2) NumInt=NumInt*(h1/3.)*(h2/3.)*(h3/3.)
If (IntRule==3) NumInt=NumInt*(3*h1/8.)*(3*h2/8.)*(3*h3/8.)

write(*,"(A,F15.6)") "Numerical integration result:",NumInt

contains

!Function to be integrated
real*8 function f(x1,x2,x3)
real*8,intent(in)::x1,x2,x3

f=(x1+x2+x3)*Sin(x1+x2+x3)

end function f

real*8 function weight(indx,n)
integer,intent(in)::indx,n
if (IntRule==1) then
    if (indx==1 .or. indx==n) then
        weight=1.
    else
        weight=2.
    endif
elseif (IntRule==2) then
    if (indx==1 .or. indx==n) then
        weight=1.
    elseif (mod(indx,2)==0) then
        weight=4.
    else
        weight=2.
    endif
elseif (IntRule==3) then
    if (indx==1 .or. indx==n) then
        weight=1.
    elseif (mod(indx-1,3)==0) then
        weight=2.
    else
```

```
        weight=3.
    endif
endif
end function weight
end
```

Εκτελώντας τον παραπάνω κώδικα για του τρεις κανόνες ολοκλήρωσης Newton-Cotes και αυξάνοντας διαδοχικά τα υποδιαστήματα προκύπτουν τα παρακάτω:

Υποδιαστήματα	Τραπέζιο	$1^{ος}$ Simpson	$2^{ος}$ Simpson
$1\times1\times1$	-24.352	-19.816	1.7020
$2\times2\times2$	0.0000	13.191	12.835
$5\times5\times5$	10.421	12.579	12.572
$10\times10\times10$	12.025	12.567	12.567
$100\times100\times100$	12.561	12.566	12.566

Άσκηση 3

Να αποδειχθεί ο αριθμητικός τύπος ολοκλήρωσης:

$$I=\int_a^b f(x)dx \cong \frac{(b-a)}{24}\left(f_0+3\sum_{i=1,4,7}[f_i+f_{i+1}]+2\sum_{i=3,6} f_i+f_9\right)$$

Απάντηση:

Σπάζοντας το παραπάνω ολοκλήρωμα σε τρία μέρη προκύπτει:

$$I=\int_a^b f(x)dx=\int_{x_0}^{x_3} f(x)dx+\int_{x_3}^{x_6} f(x)dx+\int_{x_6}^{x_9} f(x)dx, \quad x_i=a+\left(\frac{b-a}{9}\right)i=a+hi$$

Στην συνέχεια για να αποδειχθεί ο αριθμητικός τύπος ολοκλήρωσης κάθε ένα από τα τρία ολοκληρώματα αντικαθιστάται με τον $2^{ο}$ κανόνα Simpson ως:

$$I=\frac{3h}{8}(f_0+3f_1+3f_2+f_3)+\frac{3h}{8}(f_3+3f_4+3f_5+f_6)+\frac{3h}{8}(f_6+3f_7+3f_8+f_9)=$$

$$=\frac{3h}{8}(f_0+3f_1+3f_2+2f_3+3f_4+3f_5+2f_6+3f_7+3f_8+f_9)=$$

$$=\frac{3}{8}\left(\frac{b-a}{9}\right)(f_0+3f_1+3f_2+2f_3+3f_4+3f_5+2f_6+3f_7+3f_8+f_9)=$$

$$=\frac{b-a}{24}\left(f_0+3\sum_{i=1,4,7}[f_i+f_{i+1}]+2\sum_{i=3,6} f_i+f_9\right)$$

Άσκηση 4

Η αριθμητική ολοκλήρωση $I_4 = h\frac{2}{45}(7f_0 + 32f_1 + 12f_2 + 32f_3 + 7f_4)$ βασίζεται στην έκφραση πρόδρομης παρεμβολής Newton, με σφάλμα αποκοπής:

$$R = h^6\alpha(\alpha-1)(\alpha-2)(\alpha-3)(\alpha-4)(\alpha-5)\frac{f^{(6)}}{6!}$$

Με βάση το παραπάνω αποδείξτε ότι το σφάλμα ολοκλήρωσης είναι $\varepsilon = \frac{8h^7}{945}f^{(6)}$.

Απάντηση:

$$\varepsilon = h\int_0^4 h^6\alpha(\alpha-1)(\alpha-2)(\alpha-3)(\alpha-4)(\alpha-5)\frac{f^{(6)}}{6!}d\alpha =$$

$$= \frac{h^7}{720}f^{(6)}\int_0^4 \alpha(\alpha-1)(\alpha-2)(\alpha-3)(\alpha-4)(\alpha-5)d\alpha =$$

$$= \frac{h^7}{720}f^{(6)}\left(-120\alpha + 274\alpha^2 - 225\alpha^3 + 85\alpha^4 - 15\alpha^5 + \alpha^6\right)\Big|_0^4 = \frac{h^7}{720}f^{(6)}\left(-\frac{128}{21}\right) = -\frac{8h^7}{945}f^{(6)}$$

<u>Άσκηση 5</u>

Να υπολογισθούν με ολοκλήρωση Gauss η τιμή της συνάρτησης Gamma
$\Gamma(n+1) = \int_0^\infty x^n e^{-x}dx = n!$ (επιλέξτε το n)
Τα αποτελέσματα να έχουν ακρίβεια τεσσάρων σημαντικών ψηφίων. Να αποδειχθεί αριθμητικά η αναγωγική σχέση των συναρτήσεων Gamma $\Gamma(n+1) = n\Gamma(n) = n!$

Απάντηση:

Για τον υπολογισμό της συνάρτησης Gamma επιλέγεται να εφαρμοστεί αριθμητική ολοκλήρωση Gauss-Laguerre με 4 σημεία ολοκλήρωσης. Οι ρίζες και οι αντίστοιχοι συντελεστές βαρύτητας από τις σημειώσεις του μαθήματος με ακρίβεια 4 σημαντικών ψηφίων είναι:

i	x_i	w_i
1	0.3225	0.6032
2	1.746	0.3574
3	4.537	0.03889
4	9.395	0.0005393

<u>Για n=0 προκύπτει:</u>

$$\Gamma(1)=\int_0^{\infty}x^0e^{-x}dx\cong\sum_{i=1}^{4}x_i^0w_i=\sum_{i=1}^{4}w_i=0.6032+0.3574+0.03889+0.0005393=1.000$$

<u>Για n=1 προκύπτει:</u>

$$\Gamma(2)=\int_0^{\infty}xe^{-x}dx\cong\sum_{i=1}^{4}x_iw_i=$$
$$=0.3225\cdot0.6032+1.746\cdot0.3574+4.537\cdot0.03889+9.395\cdot0.0005393=$$
$$=0.1945+0.6240+0.1764+0.005067=1.000$$

<u>Για n=2 προκύπτει:</u>

$$\Gamma(3)=\int_0^{\infty}x^2e^{-x}dx\cong\sum_{i=1}^{4}x_i^2w_i=$$
$$=0.3225^2\cdot0.6032+1.746^2\cdot0.3574+4.537^2\cdot0.03889+9.395^2\cdot0.0005393=$$
$$=0.1040\cdot0.6032+3.049\cdot0.3574+20.58\cdot0.03889+88.27\cdot0.0005393=$$
$$=0.06273+1.090+0.8004+0.04760=2.001$$

<u>Για n=3 προκύπτει:</u>

$$\Gamma(4)=\int_0^{\infty}x^3e^{-x}dx\cong\sum_{i=1}^{4}x_i^3w_i=$$
$$=0.3225^3\cdot0.6032+1.746^3\cdot0.3574+4.537^3\cdot0.03889+9.395^3\cdot0.0005393=$$
$$=0.03354\cdot0.6032+5.323\cdot0.3574+93.39\cdot0.03889+829.3\cdot0.0005393=$$
$$=0.02023+1.902+3.632+0.4472=6.001$$

Από τα παραπάνω αριθμητικά αποτελέσματα φαίνεται ότι ισχύει σε τρία σημαντικά ψηφία $\Gamma(4)=3\Gamma(3)=3\left(2\Gamma(2)\right)=6\Gamma(1)=3!$ και άρα αποδεικνύεται αριθμητικά ο αναγωγικός τύπος της συνάρτησης Gamma $\Gamma(n+1)=n\Gamma(n)=n!$.

Εναλλακτικά μπορεί να εφαρμοστεί ο μετασχηματισμός:

$$s=\frac{x-1}{x+1}\qquad ds=\frac{2dx}{(x+1)^2}\qquad\Rightarrow\qquad x=\frac{1+s}{1-s}\qquad dx=\frac{2ds}{(1-s)^2}$$

$$\Gamma(n+1)=\int_0^{\infty}x^ne^{-x}dx=\int_{-1}^{1}\left(\frac{1+s}{1-s}\right)^n\exp\left[-\left(\frac{1+s}{1-s}\right)\right]\frac{2ds}{(1-s)^2}=\ldots$$

<u>Άσκηση 6</u>

Να υπολογισθούν με ολοκλήρωση Gauss τα ολοκληρώματα:

$$\alpha)\ I_1 = \int_1^5 \frac{dy}{1+2y^2}, \qquad \beta)\ I_2 = \int_{-2}^{3} y^2 dy, \qquad \gamma)\ I_3 = \int_{-1}^{1}\int_{-1}^{1} \frac{y^2}{\sqrt{1-x^2}} dy dx$$

Απάντηση:

Ο υπολογισμός των ζητούμενων ολοκληρωμάτων με Mathematica γίνεται ως:

```
NumberForm[NIntegrate[1/(1+y^2),{y,1,5}],5]
NumberForm[NIntegrate[y^2,{y,-2,3}],5]
NumberForm[NIntegrate[y^2/Sqrt[1-x^2],{y,-1,1},{x,-1,1}],5]
```

Από τα παραπάνω προκύπτουν $I_1 = 0.33587$, $I_2 = 11.667$ και $I_3 = 2.0944$.

Ο αντίστοιχος κώδικας σε Fortran που υπολογίζει οποιοδήποτε μονό ολοκλήρωμα μέσω αριθμητικής ολοκλήρωσης Gauss-Legendre με 4 σημεία για οποιοδήποτε αριθμό υποδιαστημάτων δίνεται παρακάτω:

```
program GaussLegendre1D
implicit none
integer,parameter::n=4
integer::i,j,m
real*8::a,b,h,x,w,xa,xb,NumInt,rx(n),rw(n)

!Definition of Gauss-Legendre roots and weigths
rx(1)=-0.861136311594d0; rw(1)=0.347854845137d0
rx(2)=-0.339981043585d0; rw(2)=0.652145154863d0
rx(3)= 0.339981043585d0; rw(3)=0.652145154863d0
rx(4)= 0.861136311594d0; rw(4)=0.347854845137d0

!Definition of numerical integration
a=1.  !Lower integral bound
b=5.  !Upper integral bound
m=2         !Number of integration intervals

h=(b-a)/Dble(m)
NumInt=0.
do i=1,m
    xa=a+(i-1)*h
    xb=xa+h
    do j=1,n
        x=(rx(j)*(xb-xa)+(xb+xa))/2.
        w=rw(j)*(xb-xa)/2.
        NumInt=NumInt+f(x)*w
    enddo
enddo

write(*,"(A,F15.6)") "Numerical integration result:",NumInt

contains

!Function to be integrated
real*8 function f(x)
real*8,intent(in)::x

f=1/(1.+2.*x**2.)

end function f

end
```

Για τον υπολογισμό των μονών ολοκληρωμάτων η μόνη αλλαγή που χρειάζεται στον παραπάνω κώδικα είναι ο ορισμός της συνάρτησης f η οποία ολοκληρώνεται. Αντίστοιχα το διπλό ολοκλήρωμα μπορεί να σπάσει σε δύο μονά ως:

$$I_3 = \int_{-1}^{1}\int_{-1}^{1} \frac{y^2}{\sqrt{1-x^2}} dy dx = I_{31} I_{32} = \int_{-1}^{1} \frac{dx}{\sqrt{1-x^2}} \int_{-1}^{1} y^2 dy$$

Εκτελώντας τον παραπάνω κώδικα προκύπτουν:

Υποδιαστήματα	I_1	I_2	I_{31}	I_{32}	$I_3 = I_{31} I_{32}$

1	0.33567	11.667	2.7554	0.66667	1.8369
2	0.33587	11.667	2.8681	0.66667	1.9121
10	0.33587	11.667	3.0191	0.66667	2.0128

Προφανώς το διπλό ολοκλήρωμα I_3 μπορεί να υπολογιστεί αυτούσιο χωρίς να σπάσει σε δύο μονά ολοκληρώματα. Από την μορφή του ολοκληρώματος επιλέγεται να εφαρμοσθεί ολοκλήρωση Gauss-Chebyshev για την διεύθυνση x και ολοκλήρωση Gauss-Legendre για την y οπότε προκύπτει:

$$I_3 = \int_{-1}^{1}\int_{-1}^{1} \frac{y^2}{\sqrt{1-x^2}} dydx = \sum_{i=1}^{n_x}\sum_{j=1}^{n_y} y_j^2 w_i w_j$$

Ο κώδικας σε Fortran που εκτελεί τον παραπάνω υπολογισμό είναι:

```
program GaussLegendreChebyshev2D
implicit none
integer,parameter::nl=4,nc=4
integer::i,j
real*8::NumInt,xleg(nl),wleg(nl),xche(nc),wche(nc)

!Definition of Gauss-Legendre roots and weigths
xleg(1)=-0.861136311594d0; wleg(1)=0.347854845137d0
xleg(2)=-0.339981043585d0; wleg(2)=0.652145154863d0
xleg(3)= 0.339981043585d0; wleg(3)=0.652145154863d0
xleg(4)= 0.861136311594d0; wleg(4)=0.347854845137d0
!Definition of Gauss-Chebyshev roots and weigths
xche(1)=-0.923879532511d0; wche(1)=0.785398163397d0
xche(2)=-0.382683432365d0; wche(2)=0.785398163397d0
xche(3)= 0.382683432365d0; wche(3)=0.785398163397d0
xche(4)= 0.923879532511d0; wche(4)=0.785398163397d0
!Numerical integration
NumInt=0.
do i=1,nl
    do j=1,nc
    NumInt=NumInt+xleg(i)**2.*wleg(i)*wche(j)
    enddo
enddo
write(*,"(A,F15.6)") "Numerical integration result:",NumInt
end
```

Εκτελώντας τον παραπάνω κώδικα προκύπτει $I_3 = 2.0944$.

Παρατηρείται ότι η ακρίβεια υπολογισμού του παραπάνω ολοκληρώματος μόνο με 4 σημεία σε κάθε κατεύθυνση αυξάνεται ραγδαία σε σχέση με τον υπολογισμό των δύο μονών ολοκληρωμάτων που οφείλεται κατά κύριο λόγο στην επιλογή του κατάλληλου κανόνα ολοκλήρωσης σε κάθε διεύθυνση.